> Soluzioni del compito di Calcolo Numerico del 11 gennaio 2005

> **Enrico Bertolazzi**

Metodo di Simpson

```
> restart:
  with(Student[Calculus1]):
```

Funzione da integrare:

```
> f := x -> sin(x)-x*exp(-2*x) ;
```

$$f := x \rightarrow \sin(x) - x\, \mathbf{e}^{(-2\,x)}$$

Integrale esatto

```
> a,b := 1,2:
  int(f(x),x=a..b);
  esatto := evalf(%);
```

$$\cos(1) - \frac{3}{4}\mathbf{e}^{(-2)} - \cos(2) + \frac{5}{4}\mathbf{e}^{(-4)}$$

$$esatto := 0.8778422286$$

Derivata seconda della funzione

```
> ddddf := (D@@4)(f) ;
```

$$ddddf := x \rightarrow \sin(x) + 32\,\mathbf{e}^{(-2\,x)} - 16\,x\,\mathbf{e}^{(-2\,x)}$$

Possibile costante che maggiore il modulo della derivata seconda

```
> #M := 1 + 32 + 16 ;
  M := evalf(1 + 16*exp(-2)) ;
```

$$M := 3.165364531$$

Verifica grafica:

```
> plot(abs(ddddf(x)),x=a..b) ;
```



Calcolo in numero minimo stimato di intervalli

```
> err := 10^(-5) = (b-a)^5/(180*n^4)*M ;
```

$$err := \frac{1}{100000} = \frac{0.01758535851}{n^4}$$

```
> isolate(err,n^4) ;
```

$$n^4 = 1758.535851$$

```
> evalf(rhs(%)^(1/4));
```

$$6.475716189$$

Intervalli stimati: 8

```
> ApproximateInt(f(x), x=a..b, method = simpson, output = plot,
  partition = 4);
```




```
> simp := evalf(ApproximateInt(f(x), x=a..b, method = simpson,
  partition = 4));
```

$$simp := 0.8778443466$$

Verifica della stima

```
> simp - esatto ;
```

$$0.0000021180$$

Gauss sistema lineare

```
> restart:
  with(LinearAlgebra) :
```

matrice identita e vettori canonici della base usati per i conti successivi

```
> e1,e2,e3,e4 := <1,0,0,0>,<0,1,0,0>,<0,0,1,0>,<0,0,0,1>;
```

$$e1, e2, e3, e4 := \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}$$

Matrice identità

```
> ID := <e1|e2|e3|e4>;
```

$$ID := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Matrice iniziale

```
> A := Matrix([[0,  0,  -6, 4],
               [1,  2,  0, 1],
               [1,  0,  8, 2],
               [2,  4,  4, 1]]);
```

$$A := \begin{bmatrix} 0 & 0 & -6 & 4 \\ 1 & 2 & 0 & 1 \\ 1 & 0 & 8 & 2 \\ 2 & 4 & 4 & 1 \end{bmatrix}$$

```
> A0 := A :
```

Scambio la prima con la quarta riga

```
> P1   := <e4|e2|e3|e1>;
  P1A0 := P1.A0;
```

$$P1 := \begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}$$

$$P1A0 := \begin{bmatrix} 2 & 4 & 4 & 1 \\ 1 & 2 & 0 & 1 \\ 1 & 0 & 8 & 2 \\ 0 & 0 & -6 & 4 \end{bmatrix}$$

Matrice di eliminazione

```
> L1 := ID-<0,P1A0[2,1],P1A0[3,1],P1A0[4,1]>.Transpose(e1)/P1A0[1,1];
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{-1}{2} & 1 & 0 & 0 \\ \frac{-1}{2} & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Primo passo del metodo di Gauss

```
> A1 := L1.P1A0;
```

$$A1 := \begin{bmatrix} 2 & 4 & 4 & 1 \\ 0 & 0 & -2 & \frac{1}{2} \\ 0 & -2 & 6 & \frac{3}{2} \\ 0 & 0 & -6 & 4 \end{bmatrix}$$

Scambio seconda e terza riga

```
> P2   := <e1|e3|e2|e4>;
  P2A1 := P2.A1;
```

$$P2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$P2A1 := \begin{bmatrix} 2 & 4 & 4 & 1 \\ 0 & -2 & 6 & \frac{3}{2} \\ 0 & 0 & -2 & \frac{1}{2} \\ 0 & 0 & -6 & 4 \end{bmatrix}$$

Matrice di eliminazione

```
> L2 := ID-<0,0,P2A1[3,2],P2A1[4,2]>.Transpose(e2)/P2A1[2,2];
```

$$L2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Secondo Passo del metodo di Gauss

```
> A2 := L2.P2A1;
```

$$A2 := \begin{bmatrix} 2 & 4 & 4 & 1 \\ 0 & -2 & 6 & \frac{3}{2} \\ 0 & 0 & -2 & \frac{1}{2} \\ 0 & 0 & -6 & 4 \end{bmatrix}$$

Scambio terza e quarta riga

```
> P3   := <e1|e2|e4|e3>;
  P3A2 := P3.A2;
```

$$P3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

$$P3A2 := \begin{bmatrix} 2 & 4 & 4 & 1 \\ 0 & -2 & 6 & \frac{3}{2} \\ 0 & 0 & -6 & 4 \\ 0 & 0 & -2 & \frac{1}{2} \end{bmatrix}$$

Matrice di eliminazione

```
> L3 := ID-<0,0,0,P3A2[4,3]>.Transpose(e3)/P3A2[3,3];
```

$$L3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & \frac{-1}{3} & 1 \end{bmatrix}$$

Terzo passo del metodo di Gauss

```
> A3 := L3.P3A2;
```

$$A3 := \begin{bmatrix} 2 & 4 & 4 & 1 \\ 0 & -2 & 6 & \frac{3}{2} \\ 0 & 0 & -6 & 4 \\ 0 & 0 & 0 & \frac{-5}{6} \end{bmatrix}$$

Matrici P, L, U

```
> P := P3.P2.P1:
  L := P.(L3.P3.L2.P2.L1.P1)^(-1):
  U := A3:
  print(`P=`,P,`  L=`, L,`  U=`,U ) ;
```

$$P=, \begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}, \quad L=, \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{1}{2} & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{1}{2} & 0 & \frac{1}{3} & 1 \end{bmatrix}, \quad U=, \begin{bmatrix} 2 & 4 & 4 & 1 \\ 0 & -2 & 6 & \frac{3}{2} \\ 0 & 0 & -6 & 4 \\ 0 & 0 & 0 & \frac{-5}{6} \end{bmatrix}$$

Controllo

```
> R := P.A - L.U;
```

$$R := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Calcolo la soluzione del primo problema

```
> b  := A.< 1, 2, -3, -4>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  print(`b=`,b,`  Pb=`,Pb,`  z=`,z,`  x=`,x) ;
```

$$b=, \begin{bmatrix} 2 \\ 1 \\ -31 \\ -6 \end{bmatrix}, \quad Pb=, \begin{bmatrix} -6 \\ -31 \\ 2 \\ 1 \end{bmatrix}, \quad z=, \begin{bmatrix} -6 \\ -28 \\ 2 \\ \frac{10}{3} \end{bmatrix}, \quad x=, \begin{bmatrix} 1 \\ 2 \\ -3 \\ -4 \end{bmatrix}$$

Calcolo la soluzione del secondo problema

```
> b  := A.< 1, 0, -10, -0>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  print(`b=`,b, `  Pb=`,Pb, `  z=`,z, `  x=`,x) ;
```

$$b=, \begin{bmatrix} 60 \\ 1 \\ -79 \\ -38 \end{bmatrix}, \ Pb=, \begin{bmatrix} -38 \\ -79 \\ 60 \\ 1 \end{bmatrix}, \ z=, \begin{bmatrix} -38 \\ -60 \\ 60 \\ 0 \end{bmatrix}, \ x=, \begin{bmatrix} 1 \\ 0 \\ -10 \\ 0 \end{bmatrix}$$

ODE Taylor

```
> restart;
> h := 'h';
```

$$h := h$$

Funzione da integrare

```
> f := (x,y) -> x*y - exp(y) ;
```

$$f := (x, y) \to x\,y - \mathrm{e}^{y}$$

Derivata prima

```
> dy := f(x,y) ;
```

$$dy := x\,y - \mathrm{e}^{y}$$

Derivata seconda

```
> ddy := expand(diff(dy,x)+diff(dy,y)*f(x,y)) ;
```

$$ddy := y + x^2\,y - x\,\mathrm{e}^{y} - \mathrm{e}^{y}\,x\,y + (\mathrm{e}^{y})^2$$

Derivata terza

```
> dddy := expand(diff(ddy,x)+diff(ddy,y)*f(x,y)) ;
```

$$dddy := 3\,x\,y - 2\,\mathrm{e}^{y} - \mathrm{e}^{y}\,y + x^3\,y - x^2\,\mathrm{e}^{y} - 2\,x^2\,\mathrm{e}^{y}\,y + 2\,x\,(\mathrm{e}^{y})^2 - \mathrm{e}^{y}\,x^2\,y^2 + 3\,(\mathrm{e}^{y})^2\,x\,y - 2\,(\mathrm{e}^{y})^3$$

Costruzione della serie di Taylor troncata

```
> ynew := unapply( y+dy*h+ddy*h^2/2+dddy*h^3/6, x, y, h) ;
```

$$ynew := (x, y, h) \to y + (x\,y - \mathrm{e}^{y})\,h + \frac{1}{2}\left(y + x^2\,y - x\,\mathrm{e}^{y} - \mathrm{e}^{y}\,x\,y + (\mathrm{e}^{y})^2\right)h^2 + \frac{1}{6}\left(3\,x\,y - 2\,\mathrm{e}^{y} - \mathrm{e}^{y}\,y + x^3\,y - x^2\,\mathrm{e}^{y} - 2\,x^2\,\mathrm{e}^{y}\,y + 2\,x\,(\mathrm{e}^{y})^2 - \mathrm{e}^{y}\,x^2\,y^2 + 3\,(\mathrm{e}^{y})^2\,x\,y - 2\,(\mathrm{e}^{y})^3\right)h^3$$

Calcolo alcuni passi

```
> x0, y0 := 1, -1;
```

$$x0, y0 := 1, -1$$

```
> h := 0.1 ;
```

$$h := 0.1$$

```
> x1 := x0 + h ;
  y1 := evalf(ynew(x0, y0, h)) ;
```

$$x1 := 1.1$$

$$y1 := -1.146878399$$

```
> x2 := x1 + h ;
  y2 := evalf(ynew(x1, y1, h)) ;
```

$$x2 := 1.2$$

$$y2 := -1.317678751$$

```
> x3 := x2 + h ;
  y3 := ynew(x2, y2, h) ;
```

$$x3 := 1.3$$

$$y3 := -1.519024100$$

```
>
```

Spline cubica

```
> restart;
  with(LinearAlgebra):
> # Punti di interpolazione
> X0,X1,X2,X3,X4 := 0,1,2,3,4 ;
  Y0,Y1,Y2,Y3,Y4 := 0,1,0,1,0 ;
```

$$X0, X1, X2, X3, X4 := 0, 1, 2, 3, 4$$

$$Y0, Y1, Y2, Y3, Y4 := 0, 1, 0, 1, 0$$

```
> # condizioni al contorno
  S2_init  := -1 ;
  S2_final := -1 ;
```

$$S2\_init := -1$$

$$S2\_final := -1$$

```
> # Equazioni dei momenti
  # notate che h[k] = 1 costante perche i punti sono
  # equispaziati
  EQ1     := M0*(1/2) + 2*M1 + M2*(1/2) = 3*((Y2-Y1)-(Y1-Y0)) ;
  EQ2     := M1*(1/2) + 2*M2 + M3*(1/2) = 3*((Y3-Y2)-(Y2-Y1)) ;
  EQ3     := M2*(1/2) + 2*M3 + M4*(1/2) = 3*((Y4-Y3)-(Y3-Y2)) ;
  EQINIT  := M0 = S2_init ;
  EQFINAL := M4 = S2_final ;
```

$$EQ1 := \frac{1}{2} M0 + 2\, M1 + \frac{1}{2} M2 = -6$$

$$EQ2 := \frac{1}{2} M1 + 2\, M2 + \frac{1}{2} M3 = 6$$

$$EQ3 := \frac{1}{2} M2 + 2\, M3 + \frac{1}{2} M4 = -6$$

$$EQINIT := M0 = -1$$

$$EQFINAL := M4 = -1$$

```
> RES := solve({EQ1,EQ2,EQ3,EQINIT,EQFINAL},{M0,M1,M2,M3,M4}) ;
```

$$RES := \{M1 = -4, M2 = 5, M0 = -1, M4 = -1, M3 = -4\}$$

```
> Stratto := (x,x0,x1,y0,y1,m0,m1) ->
           y0 + ((y1-y0)/(x1-x0) - (x1-x0)*(m1+2*m0) / 6) * (x-x0)
              + (m0/2)*(x-x0)^2
              + ((m1-m0)/(6*(x1-x0)))*(x-x0)^3 ;
```

$$Stratto := (x, x0, x1, y0, y1, m0, m1) \rightarrow y0 + \left(\frac{y1 - y0}{x1 - x0} - \frac{1}{6} (x1 - x0)\,(m1 + 2\, m0) \right) (x - x0) + \frac{1}{2} m0\, (x - x0)^2 + \frac{(m1 - m0)\,(x - x0)^3}{6\, x1 - 6\, x0}$$

```
> # Calcolo i tratti di cubica
  S1 := expand(subs(RES,Stratto(x,X0,X1,Y0,Y1,M0,M1))) ;
```

$$S1 := 2\, x - \frac{1}{2} x^2 - \frac{1}{2} x^3$$

```
> S2 := expand(subs(RES,Stratto(x,X1,X2,Y1,Y2,M1,M2))) ;
```

$$S2 := -2 + 8\, x - \frac{13}{2} x^2 + \frac{3}{2} x^3$$

```
> S3 := expand(subs(RES,Stratto(x,X2,X3,Y2,Y3,M2,M3))) ;
```

$$S3 := \frac{23}{2} x^2 - 28\, x + 22 - \frac{3}{2} x^3$$

```
> S4 := expand(subs(RES,Stratto(x,X3,X4,Y3,Y4,M3,M4))) ;
```

$$S4 := -32 + 26\, x - \frac{13}{2} x^2 + \frac{1}{2} x^3$$

```
> S := piecewise( x < X1,             S1,
                  x < X2 and x >= X1, S2,
                  x < X3 and x >= X2, S3,
                  x >= X3,            S4) ;
```

$$S := \begin{cases} 2\, x - \frac{1}{2} x^2 - \frac{1}{2} x^3 & x < 1 \\ -2 + 8\, x - \frac{13}{2} x^2 + \frac{3}{2} x^3 & x < 2 \textbf{ and } 1 \le x \\ \frac{23}{2} x^2 - 28\, x + 22 - \frac{3}{2} x^3 & x < 3 \textbf{ and } 2 \le x \\ -32 + 26\, x - \frac{13}{2} x^2 + \frac{1}{2} x^3 & 3 \le x \end{cases}$$

```
> plot(S,x=X0..X4) ;
```


1
0,8
0,6
0,4
0,2
0
0
1
2
3
4
x

>